

Sensitivity Analysis Using Different Equivalence Scales

Goal

The LIS equivalence scale (square root of the number of household members) is just one among many possible equivalence scales. Obviously, the choice of the scale will have an impact on the measure calculated. The difference is all the more important when you are considering specific subgroups of the population that are treated differently by the different equivalence scales. In this exercise we will see how the head count ratio (HCR) changes when using the OECD equivalence scales (modified and original) with respect to the LIS equivalence scale.

Activity

Calculate the head count ratio (HCR) for the total population, the child population, and the elderly population in Finland in 2000. Use the LIS equivalence scale, the OECD modified scale, and the OECD original scale to calculate three different measures of equivalised income.

Guidelines

- This exercise utilizes three different equivalence scales:
 - LIS scale = square root of the number of persons in the household;
 - OECD modified scale = $1 + 0.5 * \text{number of other adult members} + 0.3 * \text{number of children below 14}$;
 - OECD original scale = $1 + 0.7 * \text{number of other adult members} + 0.5 * \text{number of children below 14}$.
- To calculate the head count ratio with the three equivalence scales for the three population groups:

- Calculate three different equivalised incomes:

```
gen elis = dpi / (d4^.5)
gen emoecd = dpi / (1+.5*(d4-1-d29)+.3*d29)
gen eooecd = dpi / (1+.7*(d4-1-d29)+.5*d29)
```

- Write a subprogram that generates the HCR for the three different groups (with a **foreach** loop):

```
program define einc
    foreach var in elis emoecd eooecd {
        display "Analysis using `var'"
        qui sum `var' [w=hweight*d4], de
        local povline = r(p50)*.5
        poverty `var' [aw=hweight*d4], line(`povline') h
        poverty `var' [aw=hweight*d29], line(`povline') h
        poverty `var' [aw=hweight*(num6574+numge75)], line(`povline') h
    }
end
```

- Call the subprogram: **einc**

Program

```
di "*** INCOME DISTRIBUTION II - Exercise 15 ***"

program define einc
  foreach var in elis emoecd eooecd {
    display "Analysis using `var'"
    qui sum `var' [w=hweight*d4], de
    local povline = r(p50)*.5
    poverty `var' [aw=hweight*d4], line(`povline') h
    poverty `var' [aw=hweight*d29], line(`povline') h
    poverty `var' [aw=hweight*(num6574+numge75)], line(`povline') h
  }
end

use hweight d4 d29 num6574 numge75 dpi if (!mi(dpi) & !(dpi==0)) using $fi00h,
clear

gen elis = dpi/(d4^.5)
gen emoecd = dpi/(1+.5*(d4-1-d29)+.3*d29)
gen eooecd = dpi/(1+.7*(d4-1-d29)+.5*d29)

einc
```

Results

HCR	LIS equivalence scale	OECD modified scale	OECD original scale
Total population	5.432	4.324	3.872
Children (< 14)	3.257	3.104	5.591
Elderly	8.469	4.736	1.179

Comments

- As expected, the HCR changes when using different equivalence scales. This is especially true for child and elderly poverty rates.
- The poverty rate changes substantially between different equivalence scales. Equivalised income e_y is calculated as:

$$e_y = y / d^\varepsilon$$

where y is the unequivalised disposable income (dpi in LIS terms), d is the size of the family ($d4$ in LIS terms) and ε is the equivalence elasticity. ε varies between 0 (indicating full economies of scale and no need to adjust household income with size) and 1 (no economies of scale, which indicates the need to use per capita income). Obviously, the larger the elasticity, the smaller the economies of scale assumed by the equivalence scale. HCR increases with ε for large households, whereas it decreases for small households. This is because with a higher ε (i.e., smaller economies of scale), large households will need more income to get the same standards of living, whereas smaller households need less income. The LIS equivalence scale uses $\varepsilon=0.5$; the original OECD scale corresponds to a value of $\varepsilon\approx 0.7$; and the modified OECD scale stands somewhere in the middle. As a result, when using the OECD scales, poverty will generally be higher in households with children (which tend to be larger than the average household), and much lower in households with elderly (which tend to be much smaller).